
Abstract Description : Dates are handled as numbers with formats in SAS R©software. The
SAS macro language is a text-handling language. Macro %do state-
ments require integers for their start and stop values.

Purpose : This article examines the issues of converting dates into integers for
use in macro %do loops. Three macros are provided: a template to
modify for reports, a generic calling macro function which contains a
macro %do loop and a function which returns a list of dates. Example
programs are provided which illustrate unit testing and calculations to
produce reports for simple and complex date intervals.

Audience : macro programmers

Information : Programs in this paper are available on the web:
Fehd [7, sco.Macro-Loops-With-Dates]

Keywords : do and %do statements; interval incrementing (intnx): intervals and
shift-index; month, putn, %sysevalf, %sysfunc, today, day-of-the-week
(weekday), year

Quote : The White Rabbit put on his spectacles. ’Where shall I begin,
please your Majesty?’ he asked. ’[Start] at the beginning,’ the
King said gravely, ’and go on till you come to the end: then stop.’
— Lewis Carroll, English author and recreational mathematician
(1832–1898), Alice’s Adventures in Wonderland.

Introduction

This article examines the issues of obtaining integers from date literals forOverview
use in macro %do loops. Examine these examples.

• iterative do loop in data step

• error in a macro %do loop

• math for intnx

This data step shows an iterative do loop first with integers and then inte-Data Loop
gers resolved from date literals, which are referencs to numbers.

1 DATA _Null_;
2 do date = 0 to 1;
3 put date= 8. date date.;
4 end;
5 put;
6 do date = ’3Jan1960’d to ’4Jan1960’d;
7 put date= 8. date date.;
8 end;
9 stop;
10 run;

date=0 01JAN60
date=1 02JAN60

date=2 03JAN60
date=3 04JAN60

This log shows that a macro %do loop cannot accept a reference but mustMacro Error
have a value. This is an issue in all artificial languages and is addressed in
Contributors [1, www-wiki.Call-by-Reference].

1 %DateLoop(loop_start =’1Jan1960’d
2 ,loop_stop =’7Jan1960’d)
ERROR: A character operand was found in the %EVAL function

or %IF condition where a numeric operand is required.
The condition was: &loop_start

The problems addressed in this article are how to resolve references —
date literals — so that a macro %do loop works like a data step loop.

Can we calculate the beginning and ending dates of an interval, easily?Math or Intnx?
Yes, if we remember the adjustments. Using the intnx function is easier.

%Let today = %sysevalf(’01May2013’d);
%Let day_of_week = %sysfunc(weekday(&today));
%Let this_week_begin = %eval(&today - &day_of_week + 1);
%Let this_week_end = %eval(&today - &day_of_week + 7);

this_week_begin: 19476 Sunday, April 28, 2013
today : 19479 Wednesday, May 1, 2013
this_week_end : 19482 Saturday, May 4, 2013

Information

Dates are numbers. The mathematical operations associated with datesOverview
are addition and subtraction. The macro language handles text. The prob-
lem is how to convert text references — a date literal which represents an
integer — into an integer to be used as the argument to a macro %do loop.

To obtain an integer use the today function or a date literal.Data Step
data _null_;
zero = 0;
earlier = ’21Dec2012’d;
today = today();
put zero = zero weekdate29.;
put earlier= earlier mmddyy10. ;
put today = today date9. ;
stop;
run;

zero = 0 Friday, January 1, 1960
earlier=19348 12/21/2012
today =19387 29JAN2013

In macros we use several different functions while working with dates.Macros

assignment : use %sysfunc and today functions or %sysevalf a date literal

%let today = %sysfunc(today());
%let today = %sysevalf(’01May2013’d);

intnx : get the surrounding dates of the interval by using the intnx function

*syntax: intnx(interval in (day, week, month, quarter, year)
,start-from: an integer of a date
,increment: an integer: negative zero or positive
,alignment in (begin, middle, end, same);

%let interval = week;
%let D_Begin =%sysfunc(intnx(&interval,&today,0,begin));
%let D_End =%sysfunc(intnx(&interval,&today,0,end ));

Note: in this example the third argument of intnx is zero which re-
turns the begin and end of the current time interval. Later usage in
calling programs has minus one to get the begin and end of the pre-
vious interval.

display : use the putn function with different formats to display the integer of
a date in a macro variable

%put D_Begin: &D_Begin %sysfunc(putn(&D_Begin,mmddyy10. ));
%put today : &today %sysfunc(putn(&today ,weekdate29.));
%put D_End : &D_End %sysfunc(putn(&D_End ,date9. ));

D_Begin: 19476 04/28/2013
today : 19479 Wednesday, May 1, 2013
D_End : 19482 04May2013

Fehd [4, sco.Macro-Vars-of-Date-and-Time] contains other functions and
formats used to display dates in titles and footnotes.

Macros to Handle Dates

The following sections show macros Date-Report-Zero, DateLoop and ccYY-Overview
MM. Macro function DateLoop calls Date-Report-Zero with the parameters
Date-Begin and Date-End. Macro ccYY-MM is a subroutine of DateLoop
which returns a list of dates.

The programs provided here are basic unit tests and demonstrations. For
other programs see Fehd [7, sco.Macro-Loops-With-Dates].

Each program has a common header.Header
/* name: dateloop-demo.sas

description: testing
purpose: template
/******/

options mprint;
%let today = %sysfunc(today());
%let today = %sysevalf(’01May2013’d);*unit testing;

This program is used to generate test data.Test Data
test-data.sas

1 /* name: test-data
2 description: provide data with dates
3 purpose : for use by other demo programs
4 usage:
5 %daterpt0(data = Library.TestData
6 ,var = date
7 ,...);
8 /*******/
9 DATA Library.TestData;

10 attrib EntityId length = 4
11 Date length = 8 format = weekdate17.
12 Fact length = 8;
13 do Date = 0 to ’01May2013’d;
14 EntityId = int(Date /17);
15 Fact = int(Date**3/19);
16 output;
17 end;
18 stop;
19 run;

Use this calendar when comparing dates in logs.Calendar

2013
day 1 day 2 3 4 5 6 day 7

Sun Mon Tues Wed Thurs Fri Sat
Mar 31 Apr 1 2 3 4 5 6

7 8 9 10 11 12 13
14 15 16 17 18 19 20
21 22 23 24 25 26 27
28 29 30 May 1

SGF 2 3 4
5 6 7 8 9 10 11

12 13 14 15 16 17 18
19 20 21 22 23 24 25
26 27 28 29 30 31

Macro Date-Report-Zero

This section presents a simple model — a template for a macro that canOverview
be customized — with formats and macro functions for assembling ODS
filenames.

A reporting macro needs the following parameters.Model

• data set name

• variable name

• low

• high

This example shows the data set option where with a between condition.

1 PROC Print data = sashelp.class
2 (where = ( 11 <= age <= 13) );
NOTE: There were 10 observations read from the

data set SASHELP.CLASS.
WHERE (age>=11 and age<=13);

4 PROC SQL; select *
5 from sashelp.class
6 where age between 11 and 13;
7 quit;

Dijkstra [2, utexas-trans.Why-Numbering-Starts-At-Zero] explains why zero is a goodWhy Zero?
place to start: to avoid off-by-one errors.

This simple macro shows the basics of a reporting macro subroutine.Program Date Report
Zero • ODS

• data set name

• variable

• date: begin

• date: end
daterpt0.sas

1 /* name: DateRpt0.sas
2 description: report with date-begin and -end
3 purpose: template
4 /******/
5 %MACRO daterpt0
6 (data = sashelp.class
7 ,var = age
8 ,date_begin = 11
9 ,date_end = 13

10 ,testing = 0
11 );
12 ods _all_ close;
13 ODS PDF
14 file = "demo-ods-%sysfunc(juldate7(&Date_Begin)).pdf"
15 ;
16 PROC Print data = &Data.
17 (where = (&Date_Begin <= &Var <= &Date_End) );
18 title3 "data = &Data. date.var: &Var in range "
19 "%sysfunc(putn(&Date_Begin,worddate18.)) "
20 "-- %sysfunc(putn(&Date_End ,mmddyy10. ))"
21 ;
22 run;
23 ods _all_ close;
24 ods listing;
25 %mend daterpt0;

• the print procedure is used in this example; provide your own when
modifingNotes:

• to refresh date in title1 use options dtreset;

• title and footnote statements may have multiple text strings as their
arguments;

remember to end each string with a space! →
• dates may be displayed with various formats; see Fehd [4, sco.Macro-

Vars-of-Date-and-Time] for other formats and functions; to justify text in
title or footnote see Fehd [3, sco.Macro-TextLine]

• unit tests of this macro are on the web: Fehd [7, sco.Macro-Loops-With-
Dates] See program info-in-dates.sas and Fehd [4, sco.Macro-Vars-of-
Date-and-Time] for functions and formats to use in placing date-stamps
in ODS filenames.

This program can be used to test macro daterpt0 for any previous interval.Date-Report-Zero
Demo The intnx increment value of minus one is used to provide dates of the

previous interval.
daterpt0-demo-prev-any.sas

8 %let today = %sysevalf(’01May2013’d);
9

10 *choice: previous which?;
11 %let interval = year;
12 %let interval = quarter;
13 %let interval = month;
14 %let interval = week;
15

16 %let begin =%sysfunc(intnx(&interval.,&today,-1,begin));
17 %let end =%sysfunc(intnx(&interval.,&today,-1,end ));
18

19 %put today: &today %sysfunc(putn(&today,weekdate29.));
20 %put begin: &begin %sysfunc(putn(&begin,weekdate29.));
21 %put end : &end %sysfunc(putn(&end ,weekdate29.));
22

23 %daterpt0(data =library.testdata
24 ,var =date
25 ,date_begin =&begin
26 ,date_end =&end
27 );

The log shows the beginning and ending dates of the previous week.

today: 19479 Wednesday, May 1, 2013
begin: 19469 Sunday, April 21, 2013
end : 19475 Saturday, April 27, 2013

The above program shows the calculation of the begin and end dates be-Note:
fore calling the macro in order to write the values and dates to the log.

In all later examples these calculation are in the macro call.

*report previous &interval by day;
%dateloop(loop_start = %sysfunc(intnx(&interval.,&today,-1,begin))

,loop_stop = %sysfunc(intnx(&interval.,&today,-1,end ))
,interval = day
);

Contributors [1, www-wiki.Call-by-Reference] explains the difference betweenReference or Value?
calling by reference and calling by value.

Macro DateLoop

Macro DateLoop is a function: its output is either a macro call or notesOverview
written in the log while testing. Its purpose is to encapsulate a macro %do
loop through a series of dates while generating calls to a macro reporting
subroutine.

It has parameters which support these design elements:

• %do loop

• intnx interval

• called macro, passing parameters to

• testing

%do loop : • loop-start: date-begin

• loop-stop: date-end

intnx interval : simple: year, quarter, month, week, day etc.;

complex:

• multiple: intervalN: week2 is 14 days

• shift-index: inverval.N, the default is week.1 which returns the
week starting on Sunday; interval=week.2 returns the week
starting on Monday

called macro : and passing parameters to

• MacroName: the default is put note: for testing

• MacroParms: this parameter must be enclosed in the %nrstr
function to hide the special characters equal sign and comma

• called macro — e.g.: Date-Report-Zero — parameters must be
Date-Begin and Date-End

testing : this variable is provided for self-reporting, see: Fehd [8, nesug2007.cc12]
and is related to:

• MacroName, whose default is put note:

• semicolon: if the MacroName is put note: then each macro
call is a %put statement which must end with a semicolon

see also : Fehd [9, pnwsug2009.do-which] compares data and macro do iterative,
until and while.

Program DateLoop
dateloop.sas

1 /* name: ...\SAS-site\macros\dateloop.sas
2 author: Ronald J. Fehd 2013
3 description: macro function do loop from loop-start to loop-stop
4 purpose: standardize calling of report macros for many intervals
5 NOTES: called macro must have parameters Date_Begin and Date_End
6 sas.help: Incrementing Dates and Times
7 by Using Multipliers and by Shifting Intervals
8 sas.wiki: http://www.sascommunity.org/wiki/Macro_Loops_with_Dates
9 http://www.sascommunity.org/wiki/Do_which_loop_until_or_while

10 predecessor: http://www.sascommunity.org/wiki/Macro_CallMacr
11 ***********/
12 %Macro dateloop
13 (loop_start =0 /* integer of date */
14 ,loop_stop =2 /* integer of date */
15 ,interval =day /* simple in (week month quarter year)
16 /* intnx(interval: complex: week2==14 days week.2==Monday */
17 ,MacroName =put note:
18 ,MacroParms = /*%nrstr(data=sashelp.class,var=sex)*/
19 ,semicolon =0
20 ,testing =0
21 )/des = ’site: calls reporting macro with dates’;
22 %local Format MacroCall D_Begin D_End;
23 %let Format = weekdate29.;
24 %if %scan(&MacroName,1) eq put %then %let Semicolon = 1;
25 %let Testing = %eval( &Testing or &Semicolon
26 or %sysfunc(getoption(mprint)) eq MPRINT);
27 %put &SysMacroName start: %sysfunc(putn(&loop_start,&format));
28 %put &SysMacroName stop: %sysfunc(putn(&loop_stop ,&format));
29 %let D_Begin = %sysfunc(intnx(&Interval,&loop_start,0,begin));
30

31 %do %until(&D_Begin gt &loop_stop);
32 %let D_End = %sysfunc(intnx(&Interval,&D_Begin,0,end ));
33 %let MacroCall = &MacroName(;
34 %if %length(&MacroParms) %then
35 %let MacroCall = &MacroCall.%unquote(&MacroParms,);
36 %let MacroCall=&MacroCall.date_begin=&D_Begin,date_end=&D_End);
37 %if &Testing %then %do;
38 %put &SysMacroName: begin %sysfunc(putn(&D_Begin,&format));
39 %put &SysMacroName: end %sysfunc(putn(&D_End ,&format));
40 %end;
41 %&MacroCall.
42 %if &Semicolon %then %do;
43 ;
44 %end;
45 %let D_Begin = %eval(&D_End +1);
46 %end;
47 %mend dateloop;

• unit tests of this macro are in Fehd [7, sco.Macro-Loops-With-Dates]Notes:

• parameters MacroName, MacroParms and Semicolon, and assem-
blage of MacroCall are from Fehd [5, sco.Macro-CallMacro]

• macro variable testing is explained in Fehd [8, nesug2007.cc12]

• Fehd [9, pnwsug2009.do-which] compares data and macro do iterative,
until and while

This program can be used to test macro dateloop for any previous interval,DateLoop Demos
by day. The intnx increment value of minus one is used to provide dates
of the previous interval.

dateloop-demo-prev-any.sas
14 %let today = %sysevalf(’01May2013’d);
15

16 *choice: previous which?;
17 %let interval = year;
18 %let interval = quarter;
19 %let interval = month;
20 %let interval = week;
21

22 *report previous &interval by day;
23 %dateloop(loop_start = %sysfunc(intnx(&interval.,&today,-1,begin))
24 ,loop_stop = %sysfunc(intnx(&interval.,&today,-1,end ))
25 ,interval = day
26 );

The log shows the date of the previous week.

DATELOOP start: Sunday, April 21, 2013
DATELOOP stop: Saturday, April 27, 2013

These logs show the use of the week.shift-index which can be used toPrevious Month by
Week change the date-start of each weekly report.

The log of this macro call shows the loop-start is 2013-Apr-01. The defaultinterval=week
shift-index is week.1: reports begin on Sunday, the first day of the week.
The date range for the first week report contains Monday, April 1, 2013.

11 %let today = %sysevalf(’01May2013’d);
12
13 %let interval = month;
14 %let date_start =%sysfunc(intnx(&interval.,&today,-1,begin));
15 %let date_stop =%sysfunc(intnx(&interval.,&today,-1,end ));
16
17 *report first (short) week report includes first of month;
18 %dateloop(loop_start = &date_start
19 ,loop_stop = &date_stop
20 ,interval = week
21 );
DATELOOP start: Monday, April 1, 2013
DATELOOP stop: Tuesday, April 30, 2013
DATELOOP: begin Sunday, March 31, 2013
DATELOOP: end Saturday, April 6, 2013

Here the shift-index is soft-coded as the weekday of the first. The dateinterval=week.weekday
range for the first week’s report is always the 1st through the 7th.

29 *weekly reports begin on first: ThisDay in (1 8 15 22 29);
30 %dateloop(loop_start = &date_start
31 ,loop_stop = &date_stop
32 ,interval = week.%sysfunc(weekday(&date_start))
33 );
DATELOOP start: Monday, April 1, 2013
DATELOOP stop: Tuesday, April 30, 2013
DATELOOP: begin Monday, April 1, 2013
DATELOOP: end Sunday, April 7, 2013

Macro ccyy-mm

Macro DateLoop is a function: it returns no statements, only macro calls.Overview
Macro ccyy-mm is also a function: given a date it returns that date in a spe-
cial format. This is a template. See program info-in-dates.sas in Fehd [7,
sco.Macro-Loops-With-Dates] for other date parts that may be used to generate
a list of customized date-stamped tokens.

This macro returns a single token containing a date-stamped data set name:Program ccyy-mm
libref.date-ccyy-mm.

%Macro ccyy_mm(date_begin =
,date_end =
,library = work);

%local ccyy mm;
%let ccyy = %sysfunc(year(&date_begin));
%let mm = %sysfunc(putn(%sysfunc(month(&date_begin)),z2));
&library..date&ccyy._&mm
%mend;

This program shows a data step which reads the previous twelve monthsDemo: Previous-12
of date-stamped data set names.

DATA Previous_12;
do until(EndoFile);

set %dateloop
(loop_start = %sysfunc(intnx(month,&today,-12,sameday))
,loop_stop = %sysfunc(intnx(month,&today,-1 ,end ))
,interval = month
,MacroName = ccyy_mm
) end = EndoFile;

output;
end;

stop;
run;

log

DATELOOP start: Tuesday, May 1, 2012
DATELOOP stop: Tuesday, April 30, 2013
MPRINT(CCYY_MM): work.date2012_05
MPRINT(CCYY_MM): work.date2012_06
MPRINT(CCYY_MM): work.date2012_07
...
MPRINT(CCYY_MM): work.date2013_02
MPRINT(CCYY_MM): work.date2013_03
MPRINT(CCYY_MM): work.date2013_04
42 output;
43 end;
44 stop;
45 run;
NOTE: There were 31 observations read from WORK.DATE2012_05.
NOTE: There were 30 observations read from WORK.DATE2012_06.
NOTE: There were 31 observations read from WORK.DATE2012_07.
...
NOTE: There were 31 observations read from WORK.DATE2013_02.
NOTE: There were 31 observations read from WORK.DATE2013_03.
NOTE: There were 30 observations read from WORK.DATE2013_04.
NOTE: WORK.PREVIOUS_12 has 366 observations and 5 variables.

Summary

Splitting a reporting task with dates into a reporting subroutine and a callingConclusion
routine function which standardizes the macro %do loop and calculation of
date intervals yields tools that are easy to test and reuse.

Further Reading Programs : and updates to this paper are in Fehd [7, sco.Macro-Loops-With-Dates]

Inspiration : Rhodes [13, sugi31.015] provides a report of the previous week and all weeks pre-
ceeding using formats.

Dates and Times : Morgan [12, SAS-Press-2006.Morgan-Guide-Dates-Times]. Finley [11, sugi25.084]
shows how to use year.N for fiscal year.

Functions : Fehd [5, sco.Macro-CallMacro] is the predecessor and model for the macro function
DateLoop. Fehd [3, sco.Macro-TextLine] provides methods to justify text in titles and
footnotes.

List Processing : Fehd and Carpenter [10, sco.List-Processing-Basics] recommend separating the %do
loop from the report.

Macro %Do : These papers contain tips for modifying macro loops. Fehd [6, sco.Macro-Do-Loop] is
a macro function that returns tokens in a statement; macro function DateLoop can call
a macro function that does this. Fehd [9, pnwsug2009.do-which] compares data and
macro do iterative, until and while. Woodruff and Dunn [14, sesug2010.ff01] compare
do and %do loops.

Testing : Fehd [8, nesug2007.cc12] explains the use of the macro variable testing.

